

## Assignment 5 SPARQL queries

UniProt SPARQL Endpoint: <http://sparql.uniprot.org/sparql>

**Q1: 1 POINT** How many protein records are in UniProt?

```
PREFIX up: <http://purl.uniprot.org/core/>

SELECT (COUNT (?numprotein) AS ?pcount)

WHERE
{
  ?numprotein a up:Protein .
}
```

**Answer:**



The screenshot shows the UniProt SPARQL interface. At the top is the UniProt logo and a navigation bar with 'SPARQL' and 'Downloads' tabs. Below this is a 'Results' section with buttons for 'Sparql XML', 'Sparql JSON', 'CSV', 'Show query', and 'Share'. The query result is displayed in a table with one column, 'pcount', and one row containing the value '360157660' with an 'xsd:int' data type.


pcount
"360157660"xsd:int

**Q2: 1 POINT** How many *Arabidopsis thaliana* protein records are in UniProt?

```
PREFIX up:<http://purl.uniprot.org/core/>
PREFIX taxon:<http://purl.uniprot.org/taxonomy/>

SELECT (COUNT(DISTINCT ?thalianaprotein) AS ?pthalianacount)
WHERE
{
  ?thalianaprotein a up:Protein .
  ?thalianaprotein up:organism taxon:3702 .
}
```

**Answer:**



The screenshot shows the UniProt SPARQL interface. At the top is the UniProt logo and a navigation bar with 'SPARQL' and 'Downloads' tabs. Below this is a 'Results' section with buttons for 'Sparql XML', 'Sparql JSON', 'CSV', 'Show query', and 'Share'. The query result is displayed in a table with one column, 'pthalianacount', and one row containing the value '136782' with an 'xsd:int' data type.

pthalianacount
"136782"xsd:int

**Q3: 1 POINT retrieve pictures of Arabidopsis thaliana from UniProt?**

PREFIX foaf: <http://xmlns.com/foaf/0.1/>

PREFIX up: <http://purl.uniprot.org/core/>

SELECT ?pictures

WHERE

```
{  
  ?taxon a up:Taxon.  
  ?taxon up:scientificName "Arabidopsis thaliana".  
  ?taxon foaf:depiction ?pictures .  
}
```

**Answer:**

SPARQL Downloads

## Results

Sparql XML

Sparql JSON

CSV

Show query

Share

pictures



<https://upload.wikimedia.org/w>



[https://upload.wikimedia.org/wikipedia/commons/thumb/6/60/Arabidopsis\\_thaliana\\_inflorescencias.jpg/800px-Arabidopsis\\_thaliana\\_inflorescencias.jpg](https://upload.wikimedia.org/wikipedia/commons/thumb/6/60/Arabidopsis_thaliana_inflorescencias.jpg/800px-Arabidopsis_thaliana_inflorescencias.jpg)


**Q4: 1 POINT: What is the description of the enzyme activity of UniProt Protein Q9SZZ8.**

```
PREFIX uniprotkb:<http://purl.uniprot.org/uniprot/>
PREFIX skos:<http://www.w3.org/2004/02/skos/core#>
PREFIX rdfs:<http://www.w3.org/2000/01/rdf-schema#>
PREFIX up:<http://purl.uniprot.org/core/>
```

```
SELECT ?enzyme_name ?enzyme_reaction ?activity
WHERE
{
  uniprotkb:Q9SZZ8 a up:Protein ;
    up:enzyme ?enzyme .

  ?enzyme skos:prefLabel ?enzyme_name .
  ?enzyme up:activity ?activity .
  ?activity rdfs:label ?enzyme_reaction .
}
```

**Answer:**



The screenshot shows the UniProt website interface. At the top, there's a blue header with the UniProt logo and navigation links: SPARQL, Downloads, Documentation/Help, and Contact. Below the header, the word "Results" is displayed. Underneath, there are buttons for "Sparql XML", "Sparql JSON", "CSV", "Show query", and "Share". The main content area displays a table with three columns: enzyme\_name, enzyme\_reaction, and activity. The first row contains the following data:

enzyme_name	enzyme_reaction	activity
"Beta-carotene 3-hydroxylase" <sup>*xsd:string</sup>	"Beta-carotene + 4 reduced ferredoxin [iron-sulfur] cluster + 2 H(+) + 2 O(2) = zeaxanthin + 4 oxidized ferredoxin [iron-sulfur] cluster + 2 H(2)O." <sup>*xsd:string</sup>	<a href="http://purl.uniprot.org/enzyme/1.14.15.24#SIPF8A63F68B2741FFE">http://purl.uniprot.org/enzyme/1.14.15.24#SIPF8A63F68B2741FFE</a>

**Q5: 1 POINT: Retrieve the proteins ids, and date of submission, for proteins that have been added to UniProt this year (HINT Google for “SPARQL FILTER by date”).**

```
PREFIX up:<http://purl.uniprot.org/core/>
PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>

SELECT ?prot_id ?date

WHERE
{
  ?protein a up:Protein .
  ?protein up:created ?date .
  BIND (SUBSTR(STR(?protein),33) AS ?prot_id) . #the prot_id starts in the 33 position
  FILTER (?date >= "2021-01-01"^^xsd:date) .
}
```

**Answer:**


prot_id	date
"A0A1H7ADE3"xsd:string	"2021-06-02"xsd:date
"A0A1V1AIL4"xsd:string	"2021-06-02"xsd:date
"A0A2Z0L603"xsd:string	"2021-06-02"xsd:date
"A0A4J5GG53"xsd:string	"2021-04-07"xsd:date
"A0A6G8SU52"xsd:string	"2021-02-10"xsd:date
"A0A6G8SU69"xsd:string	"2021-02-10"xsd:date
"A0A7C9JLR7"xsd:string	"2021-02-10"xsd:date
"A0A7C9JMZ7"xsd:string	"2021-02-10"xsd:date
"A0A7C9KUQ4"xsd:string	"2021-02-10"xsd:date
"A0A7D4HP61"xsd:string	"2021-02-10"xsd:date
"A0A7D6A5N9"xsd:string	"2021-06-02"xsd:date
"A0A7D6FMY9"xsd:string	"2021-02-10"xsd:date
"A0A7D6VKU9"xsd:string	"2021-02-10"xsd:date
"A0A7D6VKZ9"xsd:string	"2021-02-10"xsd:date
"A0A7D7EJU1"xsd:string	"2021-02-10"xsd:date
"A0A7D7HYH9"xsd:string	"2021-02-10"xsd:date
"A0A7G5HK20"xsd:string	"2021-02-10"xsd:date
"A0A7G6B4J7"xsd:string	"2021-02-10"xsd:date
"A0A7G6T9F2"xsd:string	"2021-02-10"xsd:date
"A0A7G7EDL3"xsd:string	"2021-02-10"xsd:date
"A0A7G8TLN3"xsd:string	"2021-02-10"xsd:date
"A0A7H0XTK9"xsd:string	"2021-02-10"xsd:date
"A0A7H0ZDX6"xsd:string	"2021-02-10"xsd:date
"A0A7H1SVD2"xsd:string	"2021-02-10"xsd:date
"A0A7H4JPY6"xsd:string	"2021-02-10"xsd:date

**Q6: 1 POINT** How many species are in the UniProt taxonomy?

PREFIX up: <http://purl.uniprot.org/core/>

```
SELECT (COUNT (DISTINCT ?num) AS ?num_species_taxonomy)
WHERE
{
  ?num a up:Taxon .
  ?num up:rank up:Species .
}
```

**Answer:**



SPARQL Downloads

## Results

SPARQL XML

SPARQL JSON

CSV

Show query

Share

num_species_taxonomy
"2029846"xsd:int

**Q7: 2 POINT** How many species have at least one protein record? (this might take a long time to execute, so do this one last!).

```
PREFIX up:<http://purl.uniprot.org/core/>
```

```
SELECT (COUNT(DISTINCT ?num) AS ?species_atl_oneprot)
WHERE
{
  ?protein a up:Protein .
  ?protein up:organism ?num .
  ?num a up:Taxon .
  ?num up:rank up:Species .
}
```

**Answer:**

## Results

[Sparql XML](#) [Sparql JSON](#) [CSV](#) [Show query](#) [Share](#)

species\_atl\_oneprot

"1057158"<sup>xsd:int</sup>

**Q8: 3 points:** find the AGI codes and gene names for all *Arabidopsis thaliana* proteins that have a protein function annotation description that mentions “pattern formation”.

```
PREFIX up: <http://purl.uniprot.org/core/>
PREFIX rdfs:<http://www.w3.org/2000/01/rdf-schema#>
PREFIX skos: <http://www.w3.org/2004/02/skos/core#>
```

```
SELECT DISTINCT ?AGI_name ?name
```

```
WHERE
```

```
{
  ?protein a up:Protein ;
    up:organism ?taxon_id ;
    up:encodedBy ?g ;
    up:annotation ?function_annot .
  ?taxon_id a up:Taxon ;
    up:scientificName "Arabidopsis thaliana" .

  ?g skos:prefLabel ?name .
  ?g up:locusName ?AGI_name .
  ?protein up:annotation ?annotation .
    ?annotation rdfs:comment ?f_annot .
    FILTER REGEX (?f_annot, "pattern formation", "i") .
}
```

**Answer:**

## Results

Sparql XML Sparql JSON CSV Show query Share

AGI_name	name
"At3g54220" <sup>xsd:string</sup>	"SCR" <sup>xsd:string</sup>
"At1g13980" <sup>xsd:string</sup>	"GN" <sup>xsd:string</sup>
"At5g40260" <sup>xsd:string</sup>	"SWEET8" <sup>xsd:string</sup>
"At4g21750" <sup>xsd:string</sup>	"ATML1" <sup>xsd:string</sup>
"At1g69670" <sup>xsd:string</sup>	"CUL3B" <sup>xsd:string</sup>
"At1g63700" <sup>xsd:string</sup>	"YDA" <sup>xsd:string</sup>
"At2g46710" <sup>xsd:string</sup>	"ROPGAP3" <sup>xsd:string</sup>
"At1g26830" <sup>xsd:string</sup>	"CUL3A" <sup>xsd:string</sup>
"At1g55325" <sup>xsd:string</sup>	"MED13" <sup>xsd:string</sup>
"At3g09090" <sup>xsd:string</sup>	"DEX1" <sup>xsd:string</sup>
"At4g37650" <sup>xsd:string</sup>	"SHR" <sup>xsd:string</sup>
"At5g55250" <sup>xsd:string</sup>	"IAMT1" <sup>xsd:string</sup>
"At3g02130" <sup>xsd:string</sup>	"RPK2" <sup>xsd:string</sup>
"At2g42580" <sup>xsd:string</sup>	"TTL3" <sup>xsd:string</sup>
"At1g69270" <sup>xsd:string</sup>	"RPK1" <sup>xsd:string</sup>
"At5g02010" <sup>xsd:string</sup>	"ROPGEF7" <sup>xsd:string</sup>
"At1g66470" <sup>xsd:string</sup>	"RHD6" <sup>xsd:string</sup>
"At5g37800" <sup>xsd:string</sup>	"RSL1" <sup>xsd:string</sup>
"At1g49770" <sup>xsd:string</sup>	"BHLH95" <sup>xsd:string</sup>

**SPARQL Endpoint:** <https://rdf.metanetx.org/sparql>

**Q9: 4 POINTS:** what is the MetaNetX Reaction identifier (starts with “mnxr”) for the UniProt Protein uniprotkb:Q18A79.

```
PREFIX mtnx: <https://rdf.metanetx.org/schema/>
PREFIX up: <http://purl.uniprot.org/uniprot/>
```

```
SELECT DISTINCT ?id
```

```
WHERE{
  ?pept mtnx:peptXref up:Q18A79 .
  ?cata a mtnx:CATA ;
    mtnx:pept ?pept .
  ?gpr mtnx:cata ?cata ;
    mtnx:reac ?reac .
  ?reac a mtnx:REAC ;
    mtnx:mnxr ?mnxr .
  ?mnxr rdfs:label ?id .
}
```

**Answer:**

id
"MNXR145046"
"MNXR165934"

## FEDERATED QUERY - UniProt and MetaNetX

**Q10: 5 POINTS:** What is the official Gene ID (UniProt calls this a “mnemonic”) and the MetaNetX Reaction identifier (mnxr.....) for the protein that has “Starch synthase” catalytic activity in *Clostridium difficile* (taxon 272563).

```
PREFIX mnx: <https://rdf.metanetx.org/schema/>
PREFIX uniprotkb: <http://purl.uniprot.org/uniprot/>
PREFIX up: <http://purl.uniprot.org/core/>
PREFIX taxon: <http://purl.uniprot.org/taxonomy/>

SELECT DISTINCT ?ID ?MNXID ?activity
WHERE
{
  service <http://sparql.uniprot.org/sparql> {
    ?protein a up:Protein ;
      up:organism taxon:272563 ;
      up:mnemonic ?ID ;
      up:classifiedWith ?GO .
    ?GO rdfs:label ?activity .
    filter contains(?activity, "starch synthase")
    bind (substr(str(?protein),33) as ?prot_ac)
    bind (IRI(CONCAT(uniprotkb:,:?prot_ac)) as ?uniprotRef)
  }
  service <https://rdf.metanetx.org/sparql> {
    ?pept mnx:peptXref ?uniprotRef .
    ?cata mnx:pept ?pept .
    ?gpr mnx:cata ?cata ;
      mnx:reac ?reac .
    ?reac rdfs:label ?MNXID .
  }
}
```

**Answer:**

## Results

[Sparql XML](#) [Sparql JSON](#) [CSV](#) [Show query](#) [Share](#)

ID	MNXID	activity
"GLGA_CLOD6" <a href="#">xsd:string</a>	"mnxr165934" <a href="#">xsd:string</a>	"starch synthase activity" <a href="#">xsd:string</a>
"GLGA_CLOD6" <a href="#">xsd:string</a>	"mnxr145046c3" <a href="#">xsd:string</a>	"starch synthase activity" <a href="#">xsd:string</a>